
1 Introduction

These notes are still in a preliminary form. They are being developed for lectures
I am giving at the Arizona Winter School in Tucson, March 11-15, 2006. Any
suggestions or corrections would be greatly appreciated!

Cohomology of sheaves has become a tremendously useful method in alge-
braic geometry. Often the way that cohomology is introduced is quite abstract,
and although it is possible to compute it explicitly in simple cases with the
properties that are apparent from its definiton, it can be hard to obtain de-
tailed information, or to compute these groups for more involved examples.

The purpose of these lectures is twofold: (1) to present computational meth-
ods which allow us to determine the cohomology of coherent sheaves on projec-
tive varieties (or schemes), and (2) to present many examples and applications
of their use.

Throughout these notes, S = k[x0, . . . , xn] will denote the homogeneous
coordinate ring of Pn over a field k, k not necessarily algebraically closed. If
X = V (I) ⊂ Pn, we will denote by R = S/I its homogeneous coordinate ring.

If F is a coherent sheaf on X = V (I), what do we mean when we say that we
wish to compute the cohomology of F? We generally mean one the following:

• The k-vector space Hi(X,F), or its dimension hi(X,F), or

• The R-module
Hi
∗(X,F) =

⊕

d∈Z
Hi(X,F(d)).

Since it turns out that this is sometimes not finitely generated, we might
also wish to find, for an e ∈ Z,

Hi
≥e(X,F) =

⊕

d≥e

Hi(X,F(d)).

∗Supported by the NSF.

1



As long as you are willing to accept our definitions of cohomology and coher-
ent sheaf, you will not need to know much more about these concepts. In this
case, hopefully these lectures will convince you of their importance, and make
it easier to study the more abstract theory. If you are already familiar with
coherent sheaves and their cohomology, hopefully there will be some techniques
presented here that you will find useful!

Many of the methods presented here were done by Serre [12] in the 1950’s.
David Eisenbud has a nice chapter in Vasconcelos’ book [14] on some methods of
computing cohomology of coherent sheaves. Greg Smith’s paper on computing
global Ext [?] is very clearly written, and is also useful for computing sheaf co-
homology (since global Ext is a simple generalization of sheaf cohomology) and
of course global Ext. It also has Macaulay2 code (which has since been incorpo-
rated in Macaulay2 [5]) and several explicit examples. Many of the techniques
presented here for projective space also work over products of projective spaces
and over toric varieties (see Eisenbud, Mustata and Stillman [4]). The explicit
Bernstein-Gelfand-Gelfand correspondence and using it to compute sheaf coho-
mology comes from Eisenbud-Floystad-Schreyer [3]. See also [1] and [13]. The
technique also generalizes to allow one to compute higher direct images of co-
herent sheaves under projective morphisms. See Eisenbud, Hulek and Schreyer
[?].

2 A brief introduction to sheaf cohomology

One way to use sheaf cohomology is as a black box, using it via its properties,
such as the long exact sequence, and its values on specific sheaves. In fact,
these properties provide an axiomatic definition. In many cases, this works
quite well. Our plan in this section is to first define sheaf cohomology using the
Cech complex, present some basic properties of cohomology, do a simple example
by hand, and do a second example showing how to compute cohomology with
Macaulay2.

Our setting: Let F be a coherent sheaf on X ⊂ Pn. Let {Ui | 0 ≤ i ≤ m} be
an open affine cover of X. The standard open affine cover consists of the affine
open sets Ui = X \ V (xi) ⊂ Pn in X ⊂ Pn. For a subset λ = {λ0, λ1, . . . , λp},
let |λ| = p, and Uλ :=

⋂p
i=0 Uλi .

Definition 2.1 (Cech complex). For 0 ≤ p ≤ m, Let Cp =
⊕
|λ|=p F(Uλ), and

let
σp : Cp(F) −→ Cp+1(F)

be the natural map
σp : (fi0i1...ip) 7→ (gj0...jp+1),

where

gj0...jp+1 =
p+1∑

i=0

(−1)ifj0...ĵi...jp+1
.

The Cech complex C(F) of F is the complex

0 −→ C0(F) −→ C1(F) −→ . . . −→ Cm(F) −→ 0.

Let Hi(F) = Hi(X,F) be the i-th cohomology Hi(C) of this complex (of infinite
dimensional k-vector spaces). It is possible to check that choosing another affine
open cover would result in isomorphic cohomology groups.

With some work (which Serre did in the 1950’s (see [12]), and which we will
be in a position to prove later), one can prove the following facts:

Proposition 2.2 (Facts about cohomology). If F is a coherent sheaf on Pn,
then

(a) Hi(F) is a finite dimensional k-vector space, for all 0 ≤ i ≤ n.
(b) If d = dim suppF , then for i > d, Hi(F) = 0.
(c) For d À 0, and i > 0, Hi(F(d)) = 0.
(d) [Long exact sequence] If

0 −→ F ′ −→ F −→ F ′′ −→ 0

is a short exact sequence of coherent sheaves on Pn, then there are connecting
homomorphisms Hi(F ′′) −→ Hi+1(F ′) such that the resulting sequence

0 −→ H0(F ′) −→ H0(F) −→ H0(F ′′) −→ H1(F ′) −→ . . .

is exact.

It is not too hard from the definition to find the cohomology of all degree
twists of the structure sheaf of Pn.

Proposition 2.3. (Serre [12]) For any integer d,

H0(Pn,OPn(d)) = Sd

Hi(Pn,OPn(d)) = 0
Hn(Pn,OPn(d)) = S′−n−1−d,

where S = k[x0, . . . , xn], and the prime (’) denotes k-vector space dual.

Example 2.4. As a simple example, if X = V (f) ⊂ P3 is a cubic surface, we
can compute the cohomology of OX(d), for all d. Write O for OP3 , and consider
the following exact sequence of coherent sheaves on P3:

0 −→ O(−3)
f−→ O −→ OX −→ 0.

Tensoring with a locally free sheaf (such as O(d)) leaves this sequence exact, so

0 −→ O(d− 3) −→ O(d) −→ OX(d) −→ 0

is still exact. The long exact sequence in cohomology, combined with the pre-
vious proposition gives: 0 −→ Sd−3 −→ Sd −→ H0(OX(d)) −→ 0 is exact,
and H1(OX(d)) = 0, for all d, and H2(OX(d)) is the cokernel of the map
S′−1−d −→ S′−4−d. So H2(OX(d)) = 0, for d ≥ 0.

This technique was called “eye-balling” by David Eisenbud, in his chapter
on computing cohomology ([14]). With enough exact sequences, and other in-
formation about the sheaves, this method can be quite powerful, at least once
one has some idea of what these cohomology modules represent.

So, what are the problems with this method?
(1) It is hard to reconstruct the module structure.
(2) If not enough of the cohomology spaces are zero, it is hard to deduce the

dimensions of specific cohomology groups.

Example 2.5 (Sheaf cohomology with Macaulay2: The Fermat cubic surface).
Let’s consider a simple example of using sheaf cohomology as a black box in
Macaulay2. Consider the Fermat cubic surface in P3.

i1 : S = QQ[a..d];

i2 : R = S/(a^3+b^3+c^3+d^3);

The projective variety corresponding to this ring is
i3 : X = Proj R

o3 = X

o3 : ProjectiveVariety

The structure sheaf of X:
i4 : OO_X

o4 = OO
X

o4 : SheafOfRings

Twists of sheaves are constructed using standard notation.
i5 : OO_X(4)

1
o5 = OO (4)

X

o5 : coherent sheaf on X, free

The cohomology operator is HH.
i6 : HH^0(OO_X)

1
o6 = QQ

o6 : QQ-module, free

i7 : (rank HH^2(OO_X), rank HH^2(OO_X(-1)), rank HH^2(OO_X(-2)))

o7 = (0, 1, 4)

o7 : Sequence

i8 : rank HH^1(OO_X(-3))

o8 = 0

The module structure for a truncation of Hi
∗(F) is done using the following

syntax. In this case we are finding the truncation at degree -2.
i9 : HH^0(OO_X(>=-2))

1
o9 = R

o9 : R-module, free

The sheaf associated to a graded module:
i10 : M = ideal(a^3+b^3)

3 3
o10 = ideal(- c - d )

o10 : Ideal of R

i11 : F = sheaf M

o11 = image | -c3-d3 |

1
o11 : coherent sheaf on X, subsheaf of OO

X

i12 : HH^0(F(3))

1
o12 = QQ

o12 : QQ-module, free

We haven’t discussed the cotangent sheaf yet, but it may be obtained as follows.
i13 : cotangentSheaf X

o13 = cokernel {2} | 0 0 d c 0 a2 b2 0 |
{2} | 0 d 0 -b a2 0 c2 0 |
{2} | 0 -c -b 0 0 0 d2 a2 |
{2} | d 0 0 a b2 -c2 0 0 |
{2} | -c 0 a 0 0 -d2 0 b2 |
{2} | b a 0 0 -d2 0 0 c2 |

6
o13 : coherent sheaf on X, quotient of OO (-2)

X

i14 : HH^0(cotangentSheaf X)

o14 = 0

o14 : QQ-module

If the sheaf F is defined on a projective subvariety or subscheme X ⊂ Pn,
then F can be thought of as a sheaf i∗F on Pn, where i : X → Pn is the
inclusion map. It is easy to check that Hi(X,F) ∼= Hi(Pn, i∗F). Consequently,
throughout these notes, we may as well assume that our sheaves are defined on
Pn, and we will write Hi(F) for the cohomology group Hi(X,F) = Hi(Pn, i∗F).

3 Background

We review some basic concepts and techniques from algebra, such as free reso-
lutions, Koszul complexes, and Ext modules. We use this as an opportunity to
show how to compute with these objects in Macaulay2.

An interesting example is the rational quartic curve X ⊂ P3. This is the
image of the polynomial map P1 −→ P3 where (s, t) 7→ (s4, s3t, st3, t4).

i15 : kk = ZZ/32003;

i16 : ringP1 = kk[s,t];

i17 : S = kk[a..d];

The ideal of the rational quartic is the kernel of the following ring map.
i18 : F = map(ringP1,S,{s^4,s^3*t,s*t^3,t^4})

4 3 3 4
o18 = map(ringP1,S,{s , s t, s*t , t })

o18 : RingMap ringP1 <--- S

i19 : I = kernel F

3 2 2 2 3 2
o19 = ideal (b*c - a*d, c - b*d , a*c - b d, b - a c)

o19 : Ideal of S

i20 : R = S/I

o20 = R

o20 : QuotientRing

As expected, I has codimension 2, degree 4, and genus 0.
i21 : (codim I, degree I, genus I)

o21 = (2, 4, 0)

o21 : Sequence

As far as Macaulay2 is concerned, R is a quotient ring, a very different object
from the S-module S/I. We may form the S-module as follows.

i22 : M = S^1/I

o22 = cokernel | bc-ad c3-bd2 ac2-b2d b3-a2c |

1
o22 : S-module, quotient of S

3.1 Modules

We will mainly be interested in graded modules over R or S. First, let’s check
that the S-module M = S/I defined above is graded.

i23 : isHomogeneous M

o23 = true

Recall that the d-th twist of M =
⊕

d∈ZMd is the graded module M(d) which
is the same module M , with a new grading: M(d)e := Md+e.

i24 : S^{3}

1
o24 = S

o24 : S-module, free, degrees {-3}

Note that −3 is the degree of the generator of S(3).
In Macaulay2, to obtain the module M(3), tensor the module M with the

free module S(3). (** is used as the tensor product operator)
i25 : M ** S^{3}

o25 = cokernel {-3} | bc-ad c3-bd2 ac2-b2d b3-a2c |

1
o25 : S-module, quotient of S

The truncation M≥e of a graded module M =
⊕

d∈ZMd is defined to be the
graded module

M≥e :=
⊕

d≥e

Md.

i26 : truncM = truncate(1,M)

o26 = subquotient (| a b c d |, | bc-ad c3-bd2 ac2-b2d b3-a2c |)

1
o26 : S-module, subquotient of S

Notice that the result is a subquotient module: it is generated by the image
of the first matrix, modulo the image of the second matrix. We can obtain a
(graded) module isomorphic to this which is a quotient of a free module by using
prune.

i27 : prune truncM

o27 = cokernel {1} | 0 a 0 b c 0 0 0 0 0 |
{1} | b 0 0 0 -d a 0 c2 0 0 |
{1} | 0 0 c -d 0 0 a -d2 -bd b2 |
{1} | -d -d -d 0 0 -c -b 0 c2 -ac |

4
o27 : S-module, quotient of S

Each row corresponds to a generator of the module, in this case there are four
generators, e1, . . . , e4, each having degree 1. Each column corresponds to a
relation on these generators. For example, the first column corresponds to the
relation be2 − de4 = 0.

Finally, we will be interested in vector spaces Md of graded modules M . Use
basis(d,M) to get a map to M whose image is a k-basis of the degree d part:

i28 : basis(2,M)

o28 = | a2 ab ac ad b2 bd c2 cd d2 |

o28 : Matrix

If all you want is the dimension, you may use
i29 : numgens source basis(2,M)

o29 = 9

or
i30 : hilbertFunction(2,M)

o30 = 9

3.2 Free resolutions

Let M be a graded R-module. An exact sequence of free R-modules and maps
of degree 0 of the form

· · · → Fr → Fr−1 → . . . F1 → F0 → M → 0

is called a graded free resolution of M .
The Hilbert Syzygy theorem shows that if R = k[x0, . . . , xn] then any finitely
generated module admits a free resolution with at most n + 1 free modules.
Among all free resolutions of M there is a minimal one F with the property that
every other resolution L of M is of the form L = F ⊕ T where T is a sum of
trivial exact 0 → R(d) → R(d) → 0 complexes. The minimal free resolution F
can be recognized by the fact that there are no nonzero entries of degree zero in
any of its matrices and allows us to define several invariants of a graded module
M . If F

F : · · · → Fr → · · · → F1 → F0 → M → 0

is the minimal free resolution of M with

Fi =
bi(M)⊕

j=0

R(−ai,j(M))

we define:

• Projective dimension: pdim(M) = max{r | Fr 6= 0}
• Total Betti numbers: bi(M)

• Graded Betti numbers: bi,d(M) := |{j | ai,j(M) = d}|
• (Castelnuovo-Mumford) regularity:

reg(M) = max{ai,j(M)− i | 0 ≤ i, 1 ≤ j ≤ bi(M)}

By writing these values in terms of TorR
i (M,k), it is easy to see that these are

independent of the given minimal free resolution.
Note that over an arbitrary ring R (for example R = S/I for a homogeneous

ideal I) a module M will rarely admit a finite free resolution (even if it is graded).
In this case, pdimR(M) = ∞ and reg(M) may be finite or infinite.

Castelnuovo-Mumford regularity is often defined in terms of cohomology.
The relationship with this definition will become clear after the local duality
theorem.

Constructing a free resolution efficiently is a simple generalization of Buch-
berger’s algorithm for computing Groebner bases. For details on how to perform
the computation, and further references, see [9].

Let’s illustrate these concepts with the aid of Macaulay2 using the rational
quartic curve defined earlier.

i31 : M

o31 = cokernel | bc-ad c3-bd2 ac2-b2d b3-a2c |

1
o31 : S-module, quotient of S

i32 : C = res M

1 4 4 1
o32 = S <-- S <-- S <-- S <-- 0

0 1 2 3 4

o32 : ChainComplex

i33 : C.dd

1 4
o33 = 0 : S <----------------------------------- S : 1

| bc-ad b3-a2c ac2-b2d c3-bd2 |

4 4
1 : S <--------------------------- S : 2

{2} | -b2 -ac -bd -c2 |
{3} | c d 0 0 |
{3} | a b -c -d |
{3} | 0 0 a b |

4 1
2 : S <-------------- S : 3

{4} | d |
{4} | -c |
{4} | -b |
{4} | a |

1
3 : S <----- 0 : 4

0

o33 : ChainComplexMap

Each map is obtained using indexing:
i34 : C.dd_2

o34 = {2} | -b2 -ac -bd -c2 |
{3} | c d 0 0 |
{3} | a b -c -d |
{3} | 0 0 a b |

4 4
o34 : Matrix S <--- S

Each free module C 0, C 1, ..., in the resolution is a graded free module.
One may view the degrees (i.e. negative twists) as follows.

i35 : degrees C_1

o35 = {{2}, {3}, {3}, {3}}

o35 : List

These are the degrees of the generators of each summand. Therefore,

C1 = S(−2)⊕ S(−3)3

i36 : degrees C_2

o36 = {{4}, {4}, {4}, {4}}

o36 : List

The regularity and projective dimension of M :
i37 : regularity M

o37 = 2

i38 : pdim M

o38 = 3

The projective dimension is the length of the minimal resolution C.
i39 : length C

o39 = 3

A useful way to see the graded Betti numbers is the betti command.
i40 : betti C

o40 = total: 1 4 4 1
0: 1 . . .
1: . 1 . .
2: . 3 4 1

This says that there is one generator of the ideal in degree 2, 3 in degree 3.
There are four first syzygies, all in degree 4, and one second syzygy in degree
5. Observe that the regularity is the index of the last non-zero row in the betti
diagram, while the projective dimension is the index of the last nonzero column
(in this case 3).

For some methods of computing cohomology, we will require knowledge
about the degrees that occur in the resolution. The maximum or minimum
degree at the i spot can be obtained via:

i41 : first max degrees C_1

o41 = 3

i42 : first min degrees C_1

o42 = 2

i43 : degreeRange = (M, i) -> (
C := res M;
d := apply(degrees C_i, first);
(min d, max d));

i44 : degreeRange(M,1)

o44 = (2, 3)

o44 : Sequence

It is possible for the computation of the free resolution of M = F/J to be
too demanding in terms of either time or computer memory. In these cases,
knowing the free resolution of F/in(J), where in(J) is the submodule of initial
(lead) terms of J will also provide us with bounds. More precisely:

Proposition 3.1. Let J ⊂ F be a submodule of a free S-module F , set M =
F/J . Fix a monomial order > on F , and let in(J) ⊂ F be the submodule
generated by the lead monomials of a Groebner basis of J , under the order >.
Then for every i ≥ 0 and every degree d, the graded Betti numbers satisfy

bi,d(M) ≤ bi,d(F/in(J)).

These bounds however, are generally not sharp. As an example, consider
the normal bundle of the rational quartic in P3.

i45 : N = prune Hom(I/I^2,S^1/I);

i46 : J = presentation N;

7 15
o46 : Matrix S <--- S

i47 : betti res coker J

o47 = total: 7 15 9 1
-1: 7 12 5 .
0: . 3 4 1

i48 : inJ = leadTerm gens gb J;

7 21
o48 : Matrix S <--- S

i49 : betti res coker inJ

o49 = total: 7 21 19 5
-1: 7 12 9 2
0: . 7 8 3
1: . 2 2 .

3.3 The Koszul complex

Let W = Sn+1, with basis e0, . . . , en. The exterior p-th power ∧i(W ) is the free
S-module with basis

{eI = ei1 ∧ . . . eip | I = {0 ≤ i1 < · · · < ip ≤ n}}.

There are several equivalent ways to define the Koszul complex. Here is one
explicit version:

Definition 3.2 (Koszul complex). Let f0, . . . , fn ∈ S. The Koszul complex
K(f0, . . . , fn) is the complex

0 // S
dn+1// ∧nW

dn // . . . // ∧2W
d2 // W

d1 // S // 0 ,

where

dp(eI) :=
p∑

j=1

(−1)p+1fjeI\ij
.

One easily checks that this is a complex.

Proposition 3.3. If {f0, . . . , fn} = {x0, . . . , xn} then the Koszul complex is a
minimal free resolution of k = S/(x0, . . . xn):

0 // S(−n− 1)
dn+1 // ∧nW (−n)

dn // . . . // ∧2W (−2)
d2 // W (−1)

d1 // S // k // 0 .

For a proof, see almost any book on commutative algebra. In particular,
Matsumura [?] has a very nice presentation.

Let’s see the Koszul complex explicitly in the case of three elements:
i50 : A = kk[a,b,c];

i51 : koszul(1,vars A)

o51 = | a b c |

1 3
o51 : Matrix A <--- A

i52 : koszul(2,vars A)

o52 = {1} | -b -c 0 |
{1} | a 0 -c |
{1} | 0 a b |

3 3
o52 : Matrix A <--- A

i53 : koszul(3,vars A)

o53 = {2} | c |
{2} | -b |
{2} | a |

3 1
o53 : Matrix A <--- A

The Koszul complex is a minimal free resolution of k.
i54 : C = res coker vars A

1 3 3 1
o54 = A <-- A <-- A <-- A <-- 0

0 1 2 3 4

o54 : ChainComplex

i55 : C.dd

1 3
o55 = 0 : A <------------- A : 1

| a b c |

3 3
1 : A <-------------------- A : 2

{1} | -b -c 0 |
{1} | a 0 -c |
{1} | 0 a b |

3 1
2 : A <-------------- A : 3

{2} | c |
{2} | -b |
{2} | a |

1
3 : A <----- 0 : 4

0

o55 : ChainComplexMap

3.4 Hom, Ext, and Tor modules

Hom, Ext, and Tor can all be computed efficiently, all we need is the knowledge
of how to compute free resolutions and syzygies.

The key is the ability to compute the kernel of a homomorphism φ : M −→ N
of S-modules. If M and N are (finitely generated) free S-modules, then the
module of syzygies (i.e. the kernel) is a byproduct of computing the Groebner
basis of the submodule Im(φ) of M . If at least one of M and N is not free,
it is a good exercise to see how, using Groebner bases and syzygies, one can
determine the kernel of φ.

Exercise 3.4. Suppose that R is a Noetherian ring such that one may compute
a basis for the kernel of a map of free modules. Let M and N be finitely generated
R-modules, and let

M1
m−→ M0 −→ M −→ 0,

and
N1

n−→ N0 −→ N −→ 0

be presentations, where Mi and Nj are all free modules.

(a) Show how to compute a generating set for the kernel of a map F −→ N ,
where F is a finitely generated free R-module.

(b) A homomorphism φ : M −→ N corresponds (in a nonunique manner)
to a matrix f : M0 −→ N0 such that Im(fm) ⊂ Im(n). In terms of the three
matrices m, n, and f , find the kernel of φ.

(c) Given a complex of R-modules

M ′ β−→ M
α−→ M ′′,

compute its homology ker α
Im β .

3.4.1 Hom modules

One construction that we will use frequently is the module of homomorphisms.
For example, the S-module associated to the normal bundle (or sheaf) of X =
V (I) in Pn is HomS(I, S/I) = HomS(I/I2, S/I).

If M and N are finitely generated R-modules, HomR(M,N) may be com-
puted using syzygies in the following manner. For a free R-module F =⊕

i R(−ai),

HomR(
⊕

i

R(−ai), N) =
⊕

i

HomR(R, N(ai)) =
⊕

i

N(ai) = F ∗ ⊗R N,

and if G
φ−→ F −→ M −→ 0 is a presentation of M , then we can determine

HomR(M,N) by using the left exactness of Hom:

0 −→ HomR(M, N) −→ Hom(F, N)
φ∗−→ Hom(G, N).

Therefore HomR(M, N) is the kernel of the map F ∗⊗R N −→ G∗⊗R N , which
may be computed using syzygies (as in 3.4).

Let’s perform these steps by hand via Macaulay2 to find the module HomS(J, S/J)
where J is the ideal of three lines in P3.

i56 : use S

o56 = S

o56 : PolynomialRing

i57 : J = intersect(ideal(a,b),ideal(a,c),ideal(b,d))

o57 = ideal (a*d, b*c, a*b)

o57 : Ideal of S

i58 : phi = presentation module J

o58 = {2} | b 0 |
{2} | 0 a |
{2} | -d -c |

3 2
o58 : Matrix S <--- S

i59 : phi’ = (transpose phi) ** S^1/J

o59 = {-3} | b 0 -d |
{-3} | 0 a -c |

o59 : Matrix

i60 : H = kernel phi’

o60 = subquotient ({-2} | 0 c d 0 0 a |, {-2} | ad bc ab 0 0 0 0 0 · · ·
{-2} | d 0 0 b c 0 | {-2} | 0 0 0 ad bc ab 0 0 · · ·
{-2} | 0 0 b 0 a 0 | {-2} | 0 0 0 0 0 0 ad b · · ·

3
o60 : S-module, subquotient of S

Use trim to provide a more efficient representation of the module.
i61 : trim H

o61 = subquotient ({-2} | a 0 0 d c 0 |, {-2} | 0 0 ad 0 0 bc 0 0 · · ·
{-2} | 0 c b 0 0 d | {-2} | 0 ad 0 0 bc 0 0 a · · ·
{-2} | 0 a 0 b 0 0 | {-2} | ad 0 0 bc 0 0 ab 0 · · ·

3
o61 : S-module, subquotient of S

The same steps may be performed automatically:
i62 : N = Hom(J,S^1/J)

o62 = subquotient ({-2} | a 0 0 d c 0 |, {-2} | 0 0 ad 0 0 bc 0 0 · · ·
{-2} | 0 c b 0 0 d | {-2} | 0 ad 0 0 bc 0 0 a · · ·
{-2} | 0 a 0 b 0 0 | {-2} | ad 0 0 bc 0 0 ab 0 · · ·

3
o62 : S-module, subquotient of S

Notice that the module N is a subquotient module, generated by the columns
of the first matrix. You can see the homomorphisms from these columns. For
example, the first column corresponds to the homomorphism sending the first
generator of I, ad to a, the other generators bc and ab to 0.

This module N is an S-module, but we know that it is also an R = S/I
module. This situation is common. In Macaulay2, the corresponding R-module
is

i63 : NR = N ** R

o63 = cokernel {-1} | 0 d 0 0 0 0 0 b 0 -c 0 0 |
{-1} | 0 0 -d 0 0 0 b 0 0 0 0 0 |
{-1} | 0 0 0 c -d 0 0 0 0 0 0 a |
{-1} | c 0 0 0 0 0 0 0 0 0 a 0 |
{-1} | -d 0 0 0 0 b 0 0 0 a 0 0 |
{-1} | 0 0 c 0 b 0 0 0 a 0 0 0 |

6
o63 : R-module, quotient of R

Given an R-module M , it is also an S-module, and we would like to produce a
presentation as a quotient of free S-modules. This may be done using the right
exactness of tensor products: If

Rb φ−→ Ra −→ M −→ 0

is a presentation for the R-module M , lift the entries of the matrix φ to S in
any way, and let

Sb φ−→ Sa −→ N −→ 0

be a presentation of the cokernel of this lift of φ. Right-exactness of tensor
products now says that M ∼= N ⊗S S/I.

In our example, we may write NR as an S-module, by lifting the presentation
matrix to S, and tensoring its cokernel with the module S/I.

i64 : NS = (coker lift(presentation NR,S) ** S^1/I)

o64 = cokernel {-1} | bc-ad c3-bd2 ac2-b2d b3-a2c 0 0 0 · · ·
{-1} | 0 0 0 0 bc-ad c3-bd2 ac2-b2d · · ·
{-1} | 0 0 0 0 0 0 0 · · ·
{-1} | 0 0 0 0 0 0 0 · · ·
{-1} | 0 0 0 0 0 0 0 · · ·
{-1} | 0 0 0 0 0 0 0 · · ·

6
o64 : S-module, quotient of S

This provides a very inefficient presentation. As above, use trim to simplify the
presentation.

i65 : trim NS

o65 = cokernel {-1} | 0 d 0 0 0 0 0 b 0 -c 0 0 0 0 0 0 0 · · ·
{-1} | 0 0 -d 0 0 0 b 0 0 0 0 0 c2d 0 0 0 0 · · ·
{-1} | 0 0 0 c -d 0 0 0 0 0 0 a -d3 bd2 0 0 b2d · · ·
{-1} | c 0 0 0 0 0 0 0 0 0 a 0 0 0 -bd2 b2d 0 · · ·
{-1} | -d 0 0 0 0 b 0 0 0 a 0 0 0 0 c2d 0 0 · · ·
{-1} | 0 0 c 0 b 0 0 0 a 0 0 0 0 0 0 0 0 · · ·

6
o65 : S-module, quotient of S

3.4.2 Ext modules

Ext is computed in a similar manner to Hom. To obtain ExtiR(M,N), we find
a free resolution of M , to at least length i+1, then apply Hom(−, N), and take
cohomology.

For example, let’s compute Ext2(S/I, S) ”by hand”, using Macaulay2, where
I ⊂ S is the rational quartic considered earlier. This is (up to a degree twist)
the canonical module of R.

i66 : C = res (S^1/I);

i67 : E2 = (ker transpose C.dd_3) / (image transpose C.dd_2)

o67 = subquotient ({-4} | a b c 0 0 0 |, {-4} | -b2 c a 0 |)
{-4} | 0 0 d a b 0 | {-4} | -ac d b 0 |
{-4} | 0 d 0 0 -c a | {-4} | -bd 0 -c a |
{-4} | -d 0 0 c 0 b | {-4} | -c2 0 -d b |

4
o67 : S-module, subquotient of S

i68 : prune E2

o68 = cokernel {-3} | c -d 0 a -b |
{-3} | d 0 c b 0 |
{-3} | 0 c b 0 a |

3
o68 : S-module, quotient of S

Or, let Macaulay2 do it for you.
i69 : Ext^2(S^1/I,S)

o69 = cokernel {-3} | c -d 0 a -b |
{-3} | d 0 c b 0 |
{-3} | 0 c b 0 a |

3
o69 : S-module, quotient of S

3.4.3 Computing TorR
i (M, N)

Tor is handled in the same manner as Ext. To compute TorR
i (M, N), take a

free resolution of M , at least to length i + 1, and then apply −⊗R N , and take
homology. A key property of Tor is that TorR

i (M, N) = TorR
i (N,M), so that

we could compute this Tor by starting with a free resolution of N instead.
i70 : k = coker vars S

o70 = cokernel | a b c d |

1
o70 : S-module, quotient of S

i71 : prune Tor_2(S^1/I,k)

o71 = cokernel {4} | 0 0 0 d 0 0 0 c 0 0 0 b 0 0 0 a |
{4} | 0 0 d 0 0 0 c 0 0 0 b 0 0 0 a 0 |
{4} | 0 d 0 0 0 c 0 0 0 b 0 0 0 a 0 0 |
{4} | d 0 0 0 c 0 0 0 b 0 0 0 a 0 0 0 |

4
o71 : S-module, quotient of S

This is the graded vector space k4, whose 4 generators are all in degree 4.

4 Coherent sheaves and graded modules

In this section, we recall the relationship between coherent sheaves on projec-
tive space and graded S-modules. As an important exercise, we consider the
computation of the space of global sections of a coherent sheaf.

We will restrict ourselves to graded S-modules M which are either finitely
generated, or, eventually finitely generated, that is, some truncation M≥d is a
finitely generated S-module. These will correspond to coherent sheaves on Pn.
We could loosen these requirements, and deal with quasicoherent sheaves. For
simplicity we will only consider the coherent case.

Definition 4.1 (Construction of M̃). Given a graded S-module M which is
eventually finitely generated, we will associate to it a coherent sheaf M̃ of OPn-
modules. To do so, we will assume the following facts:

• The standard open sets Ui are an open affine cover of Pn, with affine
intersections Ui ∩ Uj .

• On an affine scheme U , a coherent sheaf F is completely specified by its
OU (U)-module of global sections F(U).

• To specify a sheaf on an arbitrary scheme it suffices to specify sheaves Fi

on each open set in an open cover {Ui}, and compatible isomorphisms

φij : Fi|Ui∩Uj −→ Fj |Ui∩Uj

(see Ex. II.1.23 in Hartshorne [7]).

With this in mind, we define M̃ as the sheaf on Pn obtained by glueing the
sheaves

M̃(Ui) := (M ⊗S S[x−1
i ])0,

(the 0 subscript means the subset of all elements of degree 0), via the maps

(M̃ |Ui)Ui∩Uj = (M⊗S[x−1
i ]⊗S[x−1

j ])0 −→ (M⊗S[x−1
j ]⊗S[x−1

i ])0 = (M̃ |Uj )Ui∩Uj

Our hypotheses on M imply that M̃(Ui) is a finitely generated O(Ui)-module.

Proposition 4.2. Some basic properties of this construction include

(a) S̃ = OPn .

(b) The operation˜is an exact functor from the category of eventually finitely
generated graded S-modules to the category of coherent OPn-modules.

(c) If M is an eventually f.g. graded S-module such that Md = 0 for all
d À 0, then M̃ = 0.

(d) Every coherent OPn-module is of the form M̃ , for some finitely generated
graded S-module M .

We will use part (d) as our definition of coherent sheaf on Pn.
If M and N are eventually finitely generated graded S-modules, we can

define an equivalence relation M ≡ N iff there exists a d ∈ Z such that M≥d
∼=

N≥d. Let C be the category whose objects are the equivalence classes of this
relation.

Definition 4.3. Given a coherent OPn-module, F , define

H0
∗ (Pn,F) :=

⊕

d∈Z
H0(Pn,F(d)).

This is an eventually finitely generated graded S-module. If F is the extension
by zero of a sheaf on X = V (I) ⊂ Pn, then this module is also a graded R = S/I-
module.

This operation H0
∗ (−) : CoherentSheaves(Pn) −→ C is not an exact functor

(otherwise a significant portion of these notes would not be necessary!).

Proposition 4.4. Some basic properties of this construction include

(a) H0
∗ (Pn,OPn(d)) = S(d),

(b) For every coherent O-module F ,

˜H0∗ (Pn,F) ∼= F ,

(c) If M is a graded S-module, then the natural map

M −→ H0
∗ (M̃)

is an isomorphism in all degrees d À 0.

(d) The operations ˜ and H0
∗ (−) provide an equivalence of categories between

C and the category of coherent sheaves on Pn.

Exercise 4.5. Suppose that X = Pm × Pn. X comes equipped with locally
free sheaves OX(a, b), for a, b ∈ Z. Let S = k[x0, . . . , xm, y0, . . . , yn]. S has a
bigrading: deg xi = (1, 0) and deg yj = (0, 1).

(a) Given a bigraded S-module M , construct a coherent sheaf M̃ on X.

(b) Given a coherent sheaf F on X, define a bigraded S-module as

H0
∗ (X,F) =

⊕

a,b∈Z
H0(X,F ⊗OX(a, b)).

Determine the analogues of the previous two propositions, and the equivalence
of categories, in this new situation.

(c) Now generalize this to a product of any number of projective spaces.

4.1 Global sections of a coherent sheaf

Let M be a finitely generated graded S-module. For d ≥ 0, let Jd = (xd
0, . . . , x

d
n) ⊂

S. The sheaf axiom ensures that specifying a global section of a sheaf is equiv-
alent to giving sections on each open set of some open cover, which agree
on intersections. Therefore, a global section m of M̃ is determined by an
n + 1-tuple (m0

xd
0
, . . . , mn

xd
n

) of elements of degree zero, such that mi

xd
i

= mj

xd
j

in

(M ⊗ S[x−1
i x−1

j ])0.

Exercise 4.6. Show that every such global section determines an element of
HomS(Je, M)0, for some integer e ≥ d. Conversely, show that an element of
this module defines a global section of M̃ .

Exercise 4.7. Show that there are natural maps HomS(Jd,M) −→ HomS(Je,M),
for d ≤ e. Conclude that

H0(Pn, M̃) = lim
`→∞

HomS(J`,M)0,

and that
H0
∗ (Pn, M̃) = lim

`→∞
HomS(J`,M).

Exercise 4.8. Show that, given M , there is a bound a, such that for all ` ≥ a,

H0(Pn, M̃) = HomS(J`, M)0.

Hint: First prove this for graded free modules, and then consider the free reso-
lution of M .

For example, let’s find H0(P3,OX(1)), where X ⊂ P3 is the rational quartic
curve.

i72 : M = (S^1/I) ** S^{1}

o72 = cokernel {-1} | bc-ad c3-bd2 ac2-b2d b3-a2c |

1
o72 : S-module, quotient of S

i73 : hilbertFunction(0,M)

o73 = 4

i74 : M1 = prune Hom(image vars S, M)

o74 = cokernel {0} | -c -d 0 -b -a |
{-1} | bd c2 bc-ad ac b2 |

2
o74 : S-module, quotient of S

i75 : hilbertFunction(0,M1)

o75 = 5

i76 : use S

o76 = S

o76 : PolynomialRing

i77 : M2 = prune Hom(ideal(a^2,b^2,c^2,d^2), M)

o77 = cokernel {0} | -c -d 0 -b -a |
{-1} | bd c2 bc-ad ac b2 |

2
o77 : S-module, quotient of S

This is the same module as M1.
i78 : M1 == M2

o78 = true

It turns out that in this case, the dimension of H0(OX(1)) has already been
reached. Try proving this. M and M1 define the same sheaf OX(1), and
H0(OX(1)) has dimension five.

4.2 Improving the presentation of a sheaf

If F is a coherent sheaf on Pn, there are many (i.e. an infinite number of)
graded S-modules M such that F ∼= M̃ . Some of these modules may have small
complexity: short resolutions, nice generators, etc, while others may be given
with a large number of generators.

Here is the problem we wish to address: given a graded S-module M , find
another (finitely generated) graded S-module N which is “nicer” in some way,
and for which M̃ ∼= Ñ .

The canonical choice would be the module H0
∗ (M̃). Unfortunately, this

is sometimes infinitely generated. The following proposition tells us that in
certain common circumstances, this module will be finitely generated. The
proposition is a simple corollary of local duality, which we will prove later (see
Proposition 6.5).

Proposition 4.9. Let M be a finitely generated S = k[x0, . . . , xn]-module.
(a) M = H0

∗ (M̃) if and only if pdimS(M) ≤ n− 1.
(b) H0

∗ (M̃) is a finitely generated S-module if and only if ExtnS(M, S) has
finite dimension as a k-vector space.

(c) If X = V (I) is a smooth variety of positive dimension (more generally,
a locally Cohen-Macaulay scheme of positive dimension), then H0

∗ (OX) is a
finitely generated S-module.

We can use exercise 4.8 to find this module.
For example, consider X = V (I) ⊂ Pn defined by an ideal I. The saturation

Isat of I is defined to be

Isat := (I : (x0, . . . xn)∞) = {f ∈ S | xd
i f ∈ S, for all 0 ≤ i ≤ n, and d À 0}.

Exercise 4.10. Show that Ĩ ∼= Ĩsat, and S̃/I ∼= S̃/Isat.

Similarly, if I ⊂ F is an S-submodule of a graded free S-module F , we may
define Isat ⊂ F in a similar manner (Isat := (I : (x0, . . . , xn)∞) ⊂ F ), and you

should check that Ĩ ∼= Ĩsat, and F̃/I ∼= F̃/Isat.

4.3 Operations on coherent sheaves

There is a dictionary between operations on sheaves on Pn, and graded S-
modules. We describe a few of these dictionary items here.

Proposition 4.11. Let M and N be graded S-modules.

(a) M̃ ⊗OPn Ñ ∼= M̃ ⊗S N .

(b) HomOPn (M̃, Ñ) ∼= ˜HomS(M, N).

We will see many different uses of these constructions later, including the
group operations for Pic(X), and the normal sheaf.

4.4 Fitting ideals and locally free sheaves

If F is a locally free sheaf on X of rank r, then F is the sheaf of sections of
a well-defined vector bundle. It is common practice in algebraic geometry to
abuse language and call this sheaf F a vector bundle. Similarly, if F is locally
free of rank 1, we will call F a line bundle on X. Vector bundles are among the
most important sheaves on a variety X.

Using Fitting ideals one can detect whether a sheaf M̃ on X ⊂ Pn is a vector
bundle.

Definition 4.12 (Fitting ideal). Suppose that the finitely generated R-module
M has a presentation

Ra φ−→ Rb −→ M −→ 0.

The i-th Fitting ideal, Fitti(M), is the ideal of R generated by the (b−i)×(b−i)
minors of the matrix φ.

The main properties of Fitting ideals are (see Eisenbud [2] for a more detailed
treatment):

• The Fitting ideals are independent of the given presentation of M .

• V (Fitti(M)) ⊂ X is the locus of points p ∈ X such that M is not locally
generated by i elements.

• If Fittr−1(M) = 0, then the locus of points p ∈ X where M is not locally
free of rank r is V (Fittr(M)) ⊂ X.

A projective variety or scheme X is a local complete intersection if the ideal I
of X is locally generated by codim(X) elements. This is the same as saying that
the conormal sheaf Ĩ/I2 is a vector bundle of rank codim(X). Every smooth
variety is a local complete intersection.

As an example, let’s check that the rational quartic is a local complete
intersection. Since the rational quartic has codimension two, this means that
the ideal of the quartic should be locally generated by two elements.

i79 : phi = presentation module I

o79 = {2} | ac c2 b2 bd |
{3} | 0 -b 0 -a |
{3} | -b d -a c |
{3} | -d 0 -c 0 |

4 4
o79 : Matrix S <--- S

I is generated by ≤ 4 equations everywhere. Off of X, I is the unit ideal. The
locus of points of P3 where I is not generated by ≤ 1 equations is X:

i80 : trim minors(3,phi)

2 2 2 2 3 · · ·
o80 = ideal (b*c*d - a*d , b*c - a*c*d, b c - a*b*d, a*b*c - a d, c d · · ·
o80 : Ideal of S

i81 : radical trim minors(3,phi)

3 2 2 2 3 2
o81 = ideal (b*c - a*d, c - b*d , a*c - b d, b - a c)

o81 : Ideal of S

The locus of points of X where I is not generated by ≤ 2 equations is
i82 : trim minors(2,phi)

2 2 2 2
o82 = ideal (d , c*d, b*d, a*d, c , b*c, a*c, b , a*b, a )

o82 : Ideal of S

This defines the empty set in X ⊂ P3, so I is a local complete intersection of
codimension 2. One may use the fittingIdeal routine in Macaulay2.

i83 : fittingIdeal(2,module I)

2 2 2 2
o83 = ideal (d , c*d, b*d, a*d, c , b*c, a*c, b , a*b, a )

o83 : Ideal of S

We test whether the conormal sheaf Ĩ/I2 on X is locally free of rank 2. We
first create the R-module I/I2 ∼= I ⊗S S/I:

i84 : N’ = (module I) ** R

o84 = cokernel {2} | ac c2 b2 bd |
{3} | 0 -b 0 -a |
{3} | -b d -a c |
{3} | -d 0 -c 0 |

4
o84 : R-module, quotient of R

i85 : fittingIdeal(1,N’)

o85 = 0

o85 : Ideal of R

i86 : fittingIdeal(2,N’)

2 2 2 2
o86 = ideal (d , c*d, b*d, a*d, c , a*c, b , a*b, a )

o86 : Ideal of R

Since the zero set of this ideal is the empty set in X ⊂ P3, Ĩ/I2 is locally free
of rank 2 on X.

On the other hand, the conormal sheaf of the union of two planes in P4

meeting at a point is not locally free:
i87 : S1 = kk[a..e];

i88 : I1 = intersect(ideal(a,b),ideal(c,d))

o88 = ideal (b*d, a*d, b*c, a*c)

o88 : Ideal of S1

i89 : R1 = S1/I1;

i90 : N’ = (module I1) ** R1

o90 = cokernel {2} | 0 c 0 a |
{2} | c 0 0 -b |
{2} | 0 -d a 0 |
{2} | -d 0 -b 0 |

4
o90 : R1-module, quotient of R1

i91 : fittingIdeal(1,N’)

o91 = 0

o91 : Ideal of R1

i92 : fittingIdeal(2,N’)

2 2 2 2
o92 = ideal (d , c*d, c , b , a*b, a )

o92 : Ideal of R1

Therefore, Ĩ1/I2
1 is locally free of rank 2 everywhere except at the point (0, 0, 0, 0, 1)

of intersection of the two planes.

4.5 Hilbert polynomials and the Euler characteristic

Definition 4.13 (Euler characteristic). Given a coherent sheaf F on X ⊂ Pn,
the Euler characteristic χ(F) of F is

χ(F) =
dim X∑

i=0

(−1)idimHi(F).

Exercise 4.14. Show that d 7→ χ(M̃(d)) is a polynomial function, and that
this polynomial is the Hilbert polynomial PM (d) of M . Consequently show that
χ(M̃) = PM (0).

As a simple example, the Hilbert polynomial of the coordinate ring of the
rational quartic is:

i93 : P = hilbertPolynomial(S^1/I)

o93 = - 3*P + 4*P
0 1

o93 : ProjectiveHilbertPolynomial

The default in Macaulay2 is to write the Hilbert polynomial as a sum of Hilbert
polynomials of projective spaces. The intuition here is that, numerically, the
rational quartic X is like the union of 4 lines, with three intersection points.
The actual polynomial is:

i94 : P1 = hilbertPolynomial(S^1/I, Projective=>false)

o94 = 4i + 1

o94 : QQ [i]

The euler characteristic is the value of P at 0:
i95 : (P(0), euler(S^1/I))

o95 = (1, 1)

o95 : Sequence

5 Sheaves in nature

We consider many useful examples of coherent sheaves, including locally free
sheaves, the cotangent bundle, divisors and line bundles, the canonical divisor,
and several others.

5.1 Divisors and line bundles

For now, let’s suppose that X = V (I) ⊂ Pn, is a smooth projective variety
(or, perhaps only a normal projective variety). Recall that a divisor D on X

24



determines a locally free sheaf OX(D) of rank 1 on X (i.e. a line bundle). Two
divisors are linearly equivalent if and only if their corresponding line bundles
are isomorphic.

The set of isomorphism classes of line bundles on X form a group, the Picard
group Pic(X) of X, with group operation given by

OX(D) • OX(E) := OX(D)⊗OX
OX(E)

and inverse given by

OX(D)−1 := HomOX
(OX(D),OX).

We can perform these operations effectively using proposition 4.11.

Proposition 5.1. Let R = S/I be the homogeneous coordinate ring of X. If
OX(D) = M̃ , and OX(E) = Ñ , then

(a) OX(D + E) = M̃ ⊗R N,

(b) OX(−D) = ˜HomR(M, R),

(c) OX(D − E) = ˜HomR(N,M).

It is easy to produce such line bundles in the first place, because if J ⊂ R is
the ideal of an effective divisor D, then OX(−D) = J̃ .

It should be noted that the module M only determines the divisor D up to
linear equivalence. However, it is possible to recover a divisor D such that M̃
is isomorphic to OX(D):

Proposition 5.2. If M̃ is a locally free sheaf of rank one on X, then there
exists an ideal J ⊂ R and an integer d ∈ Z such that

M̃ ∼= J̃(d)

In this case, if H = V (x0) ⊂ X is the hyperplane section, then M̃ = J̃(d) is
isomorphic to OX(dH − V (J)).

Proof. Consider the R-module M∗ = HomR(M, R). Since M is locally free,
M∗ cannot be the zero module. Choose a non-zero f ∈ M∗ of degree d. Then
f : M → R(d) is a degree 0 homomorphism, and its kernel must be supported
at the ideal (x0, . . . , xn). Therefore if J(d) ⊂ R(d) is the image of f , then
M̃ ∼= J̃(d).

Example 5.3 (Canonical sheaf as an ideal). The most important line bundle
on X is the canonical sheaf ωX . We describe how to find ωX in section 5.6 later.

For now let us simply assume that for the rational quartic curve, the canon-
ical sheaf corresponds to the R-module Ext2S(S/I, S(−4)).

i96 : KX = Ext^2(S^1/I,S^{-4}) ** R

o96 = cokernel {1} | c -d 0 a -b |
{1} | d 0 c b 0 |
{1} | 0 c b 0 a |

3
o96 : R-module, quotient of R

i97 : KXdual = Hom(KX,R)

o97 = image {-1} | -cd c2 -ad -ac bd -b2 -ab |
{-1} | c2 -bd ac b2 -ad ab a2 |
{-1} | -d2 cd -bd -ad c2 -ac -b2 |

3
o97 : R-module, submodule of R

Each (generalized) column corresponds to a homomorphism. Macaulay2 pro-
vides a mechanism to get the homomorphisms corresponding to (combinations
of) generators. For example, the first generator of H:

i98 : KXdual_{0}

o98 = {1} | 1 |
{1} | 0 |
{1} | 0 |
{1} | 0 |
{1} | 0 |
{1} | 0 |
{1} | 0 |

o98 : Matrix

The corresponding homomorphism from KX to R:
i99 : f = homomorphism KXdual_{0}

o99 = | -cd c2 -d2 |

o99 : Matrix

i100 : ker f

o100 = subquotient ({1} | c -d 0 a -b |, {1} | c -d 0 a -b |)
{1} | d 0 c b 0 | {1} | d 0 c b 0 |
{1} | 0 c b 0 a | {1} | 0 c b 0 a |

3
o100 : R-module, subquotient of R

i101 : J = ideal image f

2 2
o101 = ideal (-c*d, c , -d )

o101 : Ideal of R

i102 : degree f

o102 = {1}

o102 : List

The ideal J is supported at the point p = (1, 0, 0, 0), and has multiplicity 6.

i103 : degree(R^1/J)

o103 = 6

Therefore, KX
∼= J(1), and ωX

∼= OX(H − 6p). Since OX(H) ∼= OX(4p),
this says that ωX

∼= OX(−2p), as expected, since the canonical sheaf on P1 is
OP1(−2).

5.1.1 The degree of a line bundle on a curve

Suppose that X is a smooth projective curve, and L = M̃ is a line bundle. We
know how to compute the euler characteristic of L: if the Hilbert polynomial
of M is P (d), then χ(L) = P (0). As a result, the Riemann-Roch theorem for
curves gives us the degree of a line bundle:

Proposition 5.4.
degL = χ(L)− χ(OX)

Since the rational quartic curve is rational, the degree of the canonical bundle
should be −2. Let’s check this.

i104 : degKX = euler KX - euler(S^1/I)

o104 = -2

5.1.2 Intersection numbers on a surface

Let X ⊂ Pn be a smooth projective surface. One way to define intersection
numbers for line bundles on X is the following:

Definition 5.5. If OX(D) and OX(E) are both line bundles on X, define

D · E := χ(OX)− χ(OX(−D))− χ(OX(−E)) + χ(OX(−D − E)).

It is worth mentioning that if D and E are different codimension one irre-
ducible subvarieties of X, then D · E is the number of intersection points of D
and E, counted with multiplicity.

For the next example, let’s define a Macaulay2 function for this intersection
number, given R-modules corresponding to the two sheaves.

i105 : intersectionNumber = (M,N) ->
euler ring M - euler M - euler N + euler(M ** N)

o105 = intersectionNumber

o105 : Function

Example 5.6 (Intersection theory on a cubic surface). Let X = V (F ) ⊂ P3 be
a smooth cubic surface. There are exactly 27 lines on this surface.

i106 : use S;

The following numbers are the elementary symmetric functions of 1, 2, 3, 4, 5.
i107 : (s1,s2,s3,s4,s5) = (15,85,225,274,120);

27



Exercise 7.9 in Mile’s Reid’s book [11] shows that the following cubic surface
has all of its 27 lines defined over the rationals. (Find them!)

i108 : F = a^2*c-b^2*d + a*(s5*d^2+s3*c*d+s1*c^2) - b*(s4*d^2+s2*c*d+c^2)

2 2 2 2
o108 = a c + 15a*c - b*c - b d + 225a*c*d -

------------------------------------------
2 2

85b*c*d + 120a*d - 274b*d

o108 : S

i109 : RF = S/F;

Consider the lines L1 = V (a, b), L2 = V (c, d). Their modules are
i110 : L1 = Hom(ideal(a,b),RF)

o110 = image {-1} | a c2+bd+85cd+274d2 |
{-1} | b ac+15c2+225cd+120d2 |

2
o110 : RF-module, submodule of RF

i111 : L2 = Hom(ideal(c,d),RF)

o111 = image {-1} | c b2-120ad+274bd |
{-1} | d a2+15ac-bc+225ad-85bd |

2
o111 : RF-module, submodule of RF

L1 and L2 do not meet on X.
i112 : intersectionNumber(L1,L2)

o112 = 0

Since there are only finitely many lines on X, each line must have negative
self-intersection.

i113 : intersectionNumber(L1,L1)

o113 = -1

i114 : intersectionNumber(L2,L2)

o114 = -1

5.2 Sheaves of differentials

If An is affine n-space over k, then the module (equivalently, the sheaf) of differ-
entials Ω1

An is the free A = k[x1, . . . , xn]-module with generators dx1, . . . , dxn.
If X = V (J) ⊂ An is an affine variety, then the module of differentials Ω1

X

is the quotient of the free A/J-module generated by dx1, . . . , dxn, by elements
{df | f ∈ I}:

I/I2 −→ Ω1
An ⊗A/J −→ Ω1

X −→ 0

For Pn, the sheaf of differentials Ω1
Pn should be defined so that on each

standard open set Ui, the restriction is Ω1
Ui

defined above, and that on Ui ∩ Uj

we can glue the two resulting definitions. It is a worthwile exercise to check

that there is a very efficient description of the resulting sheaf as the kernel of
the first map d1 = (x0, . . . , xn) in the Koszul complex:

0 −→ Ω1
Pn −→ On+1

Pn (−1) d1−→ OPn −→ 0.

If X ⊂ Pn is a subvariety or subscheme, another good exercise is to check
that

IX/I2
X −→ Ω1

Pn ⊗OPn OX −→ Ω1
X −→ 0.

More explicitly, we have:

Proposition 5.7 (Cotangent sheaf). Let X = V (I) ⊂ Pn. Let R = S/I. The
cotangent sheaf of X is the sheaf associated to the homology module of

F ⊗R
dj−→ R(−1)n+1 d1−→ R,

where if j : F −→ S is the generator matrix of the ideal I, then dj : F ⊗R −→
R(−1)n+1 is the jacobian matrix of j, and the second map d1 = (x0, . . . , xn).

5.3 Differential p-forms

The sheaf of differential p-forms on X is defined to be the sheaf

Ωp
X := ∧pΩ1

X .

Let’s start with projective space again, as there is a nice presentation for
this.

Proposition 5.8. There is an exact sequence

0 −→ Ωp
Pn −→ (∧pOn+1)(−p) −→ Ωp−1 −→ 0.

Therefore, Ωp ∼= ker dp, where dp is the map of sheaves corresponding to the pth
map in the Koszul complex.

Exercise 5.9. Find the sheaf cohomology Hq(Pn,Ωp
Pn(d)), for all p, q, and d.

Given any sheaf M̃ , we may find a presentation of ∧pM̃ from the following
proposition. Usually the module that we will obtain is fairly nasty. It is an in-
teresting open problem to find (in a reasonable way) more efficient presentations
for the sheaf.

Proposition 5.10 (p-th exterior power of a sheaf). Let G −→ F −→ M −→ 0
be a presentation for the R-module M , where F and G are (graded) free R-
modules. The p-th exterior power ∧pM of M has presentation

G⊗ ∧p−1F −→ ∧pF −→ ∧pM −→ 0.

As a simple example, let’s consider the sheaf of 1-forms and 2-forms on P2.

i115 : S3 = kk[a,b,c];

i116 : C = res coker vars S3

1 3 3 1
o116 = S3 <-- S3 <-- S3 <-- S3 <-- 0

0 1 2 3 4

o116 : ChainComplex

Ω1 is the kernel of d1, therefore the image of d2, and therefore the cokernel of
d3:

i117 : Omega1 = cokernel C.dd_3

o117 = cokernel {2} | c |
{2} | -b |
{2} | a |

3
o117 : S3-module, quotient of S3

Ω2 ∼= OP2(−3). The following presentation doesn’t make this immediately ob-
vious.

i118 : exteriorPower(2,Omega1)

o118 = cokernel {4} | b c 0 |
{4} | -a 0 c |
{4} | 0 -a -b |

3
o118 : S3-module, quotient of S3

We can clean up the presentation, finding that Ω2 = O(−3):
i119 : HH^0((sheaf oo)(>=0))

1
o119 = S3

o119 : S3-module, free, degrees {3}

We will not need it here, but we should mention that there is a simple
analogue for symmetric powers:

Proposition 5.11 (p-th symmetric power of a sheaf). Let G −→ F −→
M −→ 0 be a presentation for the R-module M , where F and G are (graded)
free R-modules. The p-th symmetric power SpM of M has presentation

G⊗ Sp−1F −→ SpF −→ SpM −→ 0.

For a discussion of these presentations, see [2], Proposition A.2.2.

5.4 Example: The Hodge diamond

Given a smooth projective variety X ⊂ Pn of dimension d, the Hodge diamond is
the set of numbers hp,q(X) := hq(X, Ωp

X). There are several relationships which
follow from duality: hp,q(X) = hd−p,d−q(X), and hp,q(X) = hq,p(X). The

Hodge decomposition shows that, if k = C, the singular cohomology groups
satisfy

dim Hi(X,C) =
⊕

p+q=i

hp,q(X).

Basically, this set of numbers is a good initial picture of a variety, and contains
a wealth of information. For more information about the Hodge decomposition,
see Chapter 0 in Griffiths-Harris [6].

Example 5.12 (Hodge diamond for a cubic threefold). Let X = V (x3
0 + x3

1 +
x3

2 + x3
3 + x3

4) ⊂ P4 be the Fermat cubic threefold.
i120 : P4 = kk[x_0..x_4];

i121 : IX = ideal(x_0^3+x_1^3+x_2^3+x_3^3+x_4^3);

o121 : Ideal of P4

i122 : X = variety IX;

i123 : time matrix table(4,4,(p,q) ->
rank HH^q(exteriorPower(p,cotangentSheaf X)))

-- used 53.88 seconds

o123 = | 1 0 0 0 |
| 0 1 5 0 |
| 0 5 1 0 |
| 0 0 0 1 |

4 4
o123 : Matrix ZZ <--- ZZ

It is clear that we did too much work here, as we didn’t use any of the dualities,
or the Euler characteristic. In a project/exercise later in these notes, we will
apply the techniques for computing sheaf cohomology that we learn to find faster
methods to find this diamond.

5.5 The normal and conormal sheaves

Normal vectors at p ∈ X ⊂ Pn are tangent vectors to Pn, modulo vectors tangent
to X at p. At every point p ∈ Pn, the tangent space Tp(Pn) = Tp(X) ⊕ E,
Thus E = Tp(Pn)/Tp(X). The normal bundle is obtained by performing this
operation at all points simultaneously: If X ⊂ Pn is smooth, and if TX = (Ω1

X)∗

is the tangent sheaf, then:

0 → TX → TPn ⊗OPn OX → NX/Pn → 0.

If X is not smooth, defining the normal sheaf as the dual of the conormal
sheaf Ĩ/I2 turns out to be more useful:

Definition 5.13. Let X = V (I) ⊂ Y = V (J) ⊂ Pn. Let R = S/I be the
homogeneous coordinate ring of X. The conormal sheaf of X in Y is the sheaf
associated to the graded R-module I/(J + I2). The normal sheaf of X in Y is

NX/Y = ˜HomR(I/(J + I2), R).

Note that the normal sheaf may also be defined to be the sheaf associated
to the S-module HomS(I/J, S/I). Both the normal and conormal sheaves are
locally free if X is a local complete intersection in Y .

5.6 The canonical sheaf

If X ⊂ Pn is a smooth variety of dimension d, then the canonical bundle
ωX is the line bundle ∧dΩ1

X . A canonical divisor KX is the divisor of any
meromorphic (rational) d-form on X. Thus ωX = OX(KX).

One may relax the hypotheses about smoothness of X, and then ωX may
not be a line bundle. For now, we will not worry about these extensions.

The following proposition follows from Serre duality.

Proposition 5.14. Suppose that X ⊂ Pn is a smooth variety of codimension
c. The canonical sheaf ωX is given by

ωX = ˜Extc
S(S/I, S(−n− 1)).

Proposition 5.15 (Adjunction). If X ⊂ Y ⊂ Pn are projective smooth va-
rieties, codim(X,Y ) = c, and NX/Y is the normal bundle, then the canonical
bundle ωX is

ωX = ∧cNX/Y ⊗OY ωY .

If the codimension of X in Y is one, then NX/Y = OY (X)⊗OY
OX , and so

ωX = OY (X)⊗OY
ωY ⊗OY

OX .

5.6.1 Example: Smoothness on the Hilbert scheme

The Hilbert scheme H =
∑

p(d)Hp(d) is a scheme which parametrizes the set of
all saturated ideals I ⊂ S = k[x0, . . . , xn] having Hilbert polynomial PS/I(d) =
p(d). The Hp(d) are projective algebraic sets (schemes). The study of their local
geometry is the object of deformation theory.

Groebner bases correspond to paths on H. Given an ideal I, there is a path
(i.e. a flat deformation) {It}, such that for t 6= 0, It is the same as I, except for
rescaling the variables, and for t = 0, I0 = in(I).

Proposition 5.16. Let I ⊂ S be a homogeneous ideal, and let X ⊂ Pn be the
corresponding subscheme of projective space. Let [I] ∈ H denote the correspond-
ing point on the Hilbert scheme. Let N = NX/Pn be the normal sheaf of X in
Pn. Then

(a) The Zariski tangent space of H at [I] has dimension h0(NX/Pn).
(b) If X is a local complete intersection, then every component of H through

the point [I] has dimension at least h0(N )− h1(N ).
(c) If X = V (I) is any projective scheme, then every component of H through

the point [I] has dimension at least h0(N ) − dimExt1X(Ĩ ,OX), where this last
Ext group is global Ext.

If X is a local complete intersection, then Ext1X(Ĩ ,OX) ∼= H1(N ), so (b)
follows from (c). For a proof of this proposition, see [8].

For an example, consider the Hilbert scheme corresponding to the rational
quartic curve in P3. First

i124 : N = Hom((module I) ** R,R)

o124 = image {-2} | c -16001d b c a 0 b |
{-3} | 2d2 0 2c2 d2 2bd -cd c2 |
{-3} | bd 16001c2 ac 0 b2 -ad 0 |
{-3} | 0 ac 0 b2 0 ab a2 |

4
o124 : R-module, submodule of R

i125 : HH^1(sheaf N)

o125 = 0

o125 : kk-module

i126 : HH^0(sheaf N)

16
o126 = kk

o126 : kk-module, free

Since we know from earlier that X = V (I) is a local complete intersection (all
smooth subvarieties of Pn are), this implies that [I] is a smooth point on its
Hilbert scheme, and the dimension of that irreducible component of the Hilbert
scheme is 16. Its Groebner basis gives a path on the Hilbert scheme, to its initial
ideal

i127 : I0 = ideal leadTerm I

3 2 3
o127 = ideal (b*c, c , a*c , b )

o127 : Ideal of S

i128 : R0 = S/I0

o128 = R0

o128 : QuotientRing

The normal sheaf for the initial ideal is
i129 : N0 = Hom((module I0) ** R0, R0)

o129 = image {-2} | 0 c b 0 0 0 0 0 0 |
{-3} | 0 0 0 0 0 c2 0 0 0 |
{-3} | 0 0 0 0 c2 0 0 ac b2 |
{-3} | b 0 0 c2 0 0 ac 0 0 |

4
o129 : R0-module, submodule of R0

i130 : HH^0(sheaf N0)

20
o130 = kk

o130 : kk-module, free

Since the dimension of the Zariski tangent space at [I0] is 20 > 16, the Hilbert
scheme is singular at [I0]. Most likely, several components of this Hilbert scheme
pass through that point.

i131 : HH^1(sheaf N0)

o131 = 0

o131 : kk-module

The initial ideal is not a local complete intersection, so even though H1(N ) = 0,
it doesn’t say anything about smoothness at that point.

i132 : X0 = Proj R0

o132 = X0

o132 : ProjectiveVariety

i133 : IIX0 = sheaf((module I0) ** R0)

o133 = cokernel {2} | b2 0 ac c2 |
{3} | 0 a 0 -b |
{3} | 0 -c -b 0 |
{3} | -c 0 0 0 |

1 3
o133 : coherent sheaf on X0, quotient of OO (-2) ++ OO (-3)

X0 X0

The obstruction to smoothness sits in the global Ext vector space Ext1X(Ĩ/I2,OX).
In this case, since X is not a local complete intersection, this is not H1(N ).

i134 : Ext^1(IIX0, OO_X0)

7
o134 = kk

o134 : kk-module, free

Remember that even for local complete intersections X, H1(NX) 6= 0 does not
imply that the Hilbert scheme is singular at the point corresponding to X.

Each Hilbert scheme has a canonically defined point: the lexicographic point.
This corresponds to the monomial ideal defined by the lexicographically first
monomials with the given Hilbert polynomial. A surprising result is that this
point is smooth on the Hilbert scheme. See [10] for the statement and proof.

6 Cohomology of sheaves

Given a coherent sheaf F = M̃ , we wish to compute the S-modules

Hi
∗(F) =

⊕

d∈Z
Hi(F(d)).

To simplify notation, we soemtimes denote these modules by

Hi(M) := Hi
∗(M̃).

Lemma 6.1. Define the complex C to be:

C : 0 −→
⊕

i=0..n

M [x−1
i ] −→

⊕

i<j

M [x−1
i x−1

j ] −→ · · · −→ M [(x0 . . . xn)−1] −→ 0.

Then Hi
∗(M̃) = Hi(C).

This allows us to compute the cohomology of OPn(d), for all d:
Proposition 6.2.

Hi
∗(OPn) =





S i = 0
0 1 ≤ i ≤ n− 1

1
x0...xn

k[x−1
0 , . . . , x−1

n ] i = n

Corollary 6.3. For any integer d,

H0(Pn,OPn(d)) = Sd

Hi(Pn,OPn(d)) = 0
Hn(Pn,OPn(d)) = S′−n−1−d.

Definition 6.4. Let M be a graded S-module. Define the graded k-dual of M
to be the graded S-module M∨ :=

⊕
d∈ZHomk(M−d, k).

The operation M 7→ M∨ is an exact contravariant functor. Notice that if M
is a finitely generated graded S-module, which is not zero in high degrees, then
M∨ is not finitely generated. In the above proposition, note that Hn

∗ (OPn) =
S(−n− 1)∨.

6.1 Local duality

We now prove perhaps the most generally useful formulas for finding specific co-
homology modules. This technique doesn’t exactly give us the module structure,
but it is very useful for relating homological properties of the module M , such
as depth and projective dimension, with properties of the cohomology modules.

Theorem 6.5 (Local Duality). Let M be a graded S-module. Then
(a) The following sequence is exact.

0 −→ Extn+1
S (M,S(−n−1))∨ −→ M −→ H0

∗ (M̃) −→ Extn
S(M,S(−n−1))∨ −→ 0,

(b) For i ≥ 1,
Hi
∗(M̃) ∼= Extn−i

S (M, S(−n− 1))∨.

35



Corollary 6.6. Let M be a graded S-module. Then for i ≥ 1,

hi(Pn, M̃(d)) = dim Extn−i
S (M, S)−n−1−d,

and the dimension h0(Pn, M̃(d)) of the space of global sections of M̃(d) is equal
to

dim Md + dimExtn
S(M, S)−n−1−d − dim Extn+1

S (M, S)−n−1−d.

Proof of Theorem 6.5. Lemma 6.2 provides a proof of the theorem when M = S,
and since all of our cohomologies commute with direct sums, it also proves the
theorem when M is a graded free S-module. This proves the theorem in the
situation when pdS M = 0.

Prove the general statement by induction on the projective dimension of the
module M .

An alternate proof uses the two spectral sequences corresponding to the
double complex Kp,q = CpM̃ ⊗S Fq, where

0 −→ Fr −→ Fr−1 −→ . . . −→ F1 −→ F0 −→ M −→ 0

is a graded free resolution of M .

6.2 Cohomology by approximation

The following result of Serre generalizes exercise 4.8.

Theorem 6.7. Let M be a graded S-module, and i an integer. The for all
` ≥ reg(M)-i,

Hi
∗(M̃)≥0

∼= Exti
S(J`,M)≥0,

where J` = (x`
0, x

`
1, . . . , x

`
n).

This may be proved either by using a spectral sequence, or by noting that the
result holds for graded free modules, and then using induction on the projective
dimension of M .

7 The Bernstein-Gelfand-Gelfand correspondence

We now describe a very cool relationship between the exterior algebra and co-
homology of sheaves on Pn. This section still needs to be written. For now, see
[3, 1, 13] for details.

8 Project: The Hodge diamond

Exercise 8.1. For some varieties the approach above works fine, but for va-
rieties with more complicated ideals, the computations can bog down. In this
exercise/project you will address these concerns.

Given a smooth projective variety X (by its ideal), write a Macaulay2 routine
to compute this (d+1)×(d+1) matrix of integers, by using as little computation
as possible. Here are a few points to keep in mind:

• by duality, one doesn’t need to compute all of these numbers,

• The Euler characteristic can be used to obtain one of the numbers from
the rest (and usually this number isn’t as hard to compute).

• If M and N are modules corresponding to the same sheaf, but the depth
of M is 0, and the depth of N is at least 1, then the presentation and
resolution of N are usually nicer.

8.1 Hodge diamond of the blowup of P3 along the twisted
cubic curve

8.2

9 Example: A mystery curve

10 Example: A mystery surface

11 Example: Intersection numbers on a surface

12 Example: Rational surfaces and Castelnuovo’s
theorem

In this example, we use Castelnuovo’s theorem to investigate the rationality of
a specific smooth surface in P4. The specific surface was discovered by Decker,
Popescu and Schreyer. First, let’s recall Castelnuovo’s theorem:

Theorem 12.1 (Castelnuovo). Let X be a smooth projective surface (over char-
acteristic zero). X is rational if and only if H1(OX) = H0(ω⊗2

X ) = 0.

Construction of the surface

The following construction of X = V (I) ⊂ P4 looks like we are pulling a rabbit
out of a hat. It is possible to motivate the construction using Beilinson monads,
and while that is a very interesting story, we will concentrate on analyzing the
surface via its ideal.

i135 : S = ZZ/32003[a..e];

i136 : F1 = S^{5}

1
o136 = S

o136 : S-module, free, degrees {-5}

i137 : F2 = S^{3:4}

3
o137 = S

o137 : S-module, free, degrees {-4, -4, -4}

i138 : G = S^{15:3}

15
o138 = S

o138 : S-module, free, degrees {-3, -3, -3, -3, -3, -3, -3, -3, -3, -3 · · ·
The following produces a 4 by 18 matrix with entries random of degree 1 and
2, except for a single 3 by 3 block in the lower left.

i139 : M = matrix{{random(F1,F2++G)},
{map(F2,F2,0) | random(F2,G)}};

4 18
o139 : Matrix S <--- S

i140 : C = res coker M

4 18 29 22 10 3
o140 = S <-- S <-- S <-- S <-- S <-- S <-- 0

0 1 2 3 4 5 6

o140 : ChainComplex

i141 : betti C

o141 = total: 4 18 29 22 10 3
-5: 1 3 3 1 . .
-4: 3 15 26 15 . .
-3: . . . 6 10 3

The kernel of the 29 by 16 submatrix of C.dd 3 corresponding to the first 16
columns is an ideal generated by 15 quintics and one sextic.

i142 : F = submatrix(C.dd_3,{0..15});

29 16
o142 : Matrix S <--- S

i143 : I = ideal syz F;

o143 : Ideal of S

i144 : betti res I

o144 = total: 1 16 29 18 4
0: 1 . . . .
1: . . . . .
2: . . . . .
3: . . . . .
4: . 15 26 15 3
5: . 1 3 3 1

The surface X = V (I) ⊂ P4 is smooth of codimension 2, degree 9, and sectional
genus 6.

i145 : (codim I, degree I)

o145 = (2, 9)

o145 : Sequence

i146 : genera I

o146 = {0, 6, 8}

o146 : List

The equations of I are messy, having been constructed using random polyno-
mials, and so we don’t display the specific equations defining X.

The cohomology of the surface

First we find q = h1(OX).
i147 : rank HH^1(sheaf(S^1/I))

o147 = 0

Alternatively, use local duality directly.
i148 : hilbertFunction({0},Ext^3(S^1/I,S^{-5}))

o148 = 0

The second plurigenus P2 = dim H0(ω⊗2
X ) is a bit more interesting.

i149 : KX = Ext^2(S^1/I,S^{-5});

i150 : KX2 = KX ** KX;

i151 : betti KX2

o151 = relations : total: 36 120
2: 36 120

Computing a free resolution of the module K⊗2
X is more difficult now, mainly

due to the fact that the given presentation is not very efficient. We improve the
presentation to depth 1 in the following way.

i152 : KX2sat = coker gens saturate(image presentation KX2);

i153 : betti res KX2sat

o153 = total: 36 75 63 30 6
1: . 15 . . .
2: 36 60 63 30 6

Now we are in a position to easily compute P2. By local duality, P2 is the sum
of the following two numbers, since Ext5 is zero.

i154 : hilbertFunction({0},Ext^4(KX2sat, S^{-5}))

o154 = 0

i155 : hilbertFunction({0},KX2sat)

o155 = 0

Therefore P2 = 0, so Castelnuovo’s theorem implies that X is a rational surface.

Exercise 12.2. Find a birational map between X and P2. X must be the
blowdown at some points of the blowup of P2 at some points. Try to describe
the surface along these lines.

13 Example: Rational curves on a variety

14 Example: Rational connectivity

15 Example: Connectedness

